
C  **********************************************************************       
C                                PEDROP                                         
C  THIS PROGRAM COMPUTES THE PRESSURE DROP AGAINST THE MASS FLOW RATE           
C  OF AN UNIFORMLY HEATED NOMINAL BWR SUBCHANNEL MADE UP OF FOUR RODS           
C  ARRANGED IN A SQUARE ARRAY. THE COOLANT AT THE INLET IS SINGLE               
C  PHASE PRESSURISED WATER. THE PROGRAM IS GOOD FOR THE PRESSURE                
C  RANGE OF 5.514 TO 11.029 MPA (800 TO 1600 PSIA).                             
C  C YEE-NING CHAN 1982 MASSACHUSETTS INSTITUTE OF TECHNOLOGY.                  
C                                                                               
C  **********************************************************************       
C                                                                               
C  NOMENCLATURE AND UNITS OF INPUTS.                                            
C  P      PRESSURE(MPA)                                                         
C  DROD   ROD DIAMETER(M)                                                       
C  PITCH  ROD CENTER-TO-CENTER DISTANCE(M)                                      
C  TL     CHANNEL AXIAL LENGTH(M)                                               
C  TIN    COOLANT INLET TEMPERATURE(K)                                          
C  TSAT   SATURATION TEMPERATURE(K)                                             
C  HIN    INLET ENTHALPY (KJ/KG)                                                
C  NQ     NUMBER OF HEAT FLUXES                                                 
C  Q      HEAT FLUX(MW/M**2)                                                    
C  MASS   HIGHEST MASS FLOW RATE(KG/SEC)                                        
C  DEC    MASS FLOW RATE DECREMENTS(KG/SEC)                                     
C  MINM  LOWEST MASS FLOW RATE(KG/SEC)                                          
C  LBOT BOTTOM UNHEATED LENGTH (M)                                              
C  LTOP  TOP UNHEATED LENGTH (M)                                                
C  LEBOT BOTTOM, EQUILEVENT LENGTH TO ACCOUNT FOR FORM LOSS(M)                  
C  LETOP  TOP      ''          '          '     '   '     '                     
C  IFLAG  IFLAG>0 => NO CALCULATION OF ACCELERATION PR. DROP                    
C  JFLAG  JFLAG>0 => USE DEVIDED FACTOR IN PRESSURE RANGE                       
C                     14.7 PSI <  P  <  100 PSI                                 
C  KFLAG   KFLAG>0 => TWO PHASE MULTIPLIER IN SUBCOOLED BOILING REGION =1       
C                                                                               
C  ZSC    AXIAL POINT OF INCEPTION OF SUBCOOLED BOILING(M)                      
C  ZB     AXIAL POINT OF INCEPTION OF BULK BOILING(M)                           
C  ZV     AXIAL POINT OF INCEPTION OF SINGLE PHASE VAPOR(M)                     
      COMMON Q(30)                                                              
      CHARACTER*64 FINPUT,FOUTPUT
      REAL MASS,MDOT,MINM,LBOT,LTOP,LEBOT,LETOP                                 
      DATA C4,C5,C6,C7,C8/958.75,-0.8566,2619410.618,-4.995E10,3.403E5/         
      DATA C9,C10,C11,C12/1.0665545,1.02E-8,-2.548E-15,2589600./                
      DATA C13,C14,C15,C16/6.350E-3,-1.0582E-9,1.0764,3.625E-10/                
      DATA C17,C20,C23,C24,C26/-9.063E-17,461.7,647.3,1.3,0.3/                  
      DATA PI/3.1415926/                                                        
C     CALL LINK('UNIT5=IPEDROP,UNIT6=(OPEDROP,CREATE,TEXT),UNIT59=TTY//'        
C    1)                                                                         
      WRITE(*,888)
      READ(*,887)FINPUT
       WRITE(*,886)
       READ(*,887)FOUTPUT
       OPEN(5,  FILE=FINPUT)
       OPEN(6,  FILE=FOUTPUT,STATUS='NEW')
888    FORMAT(' INPUT DATA FILE NAME'\)
887    FORMAT(A)
886    FORMAT(' OUTPUT DATA FILE NAME'\)
C     CALL ERRSET(208,256,-1,0,0,0)                                             
      READ(5,1001) NSET                                                         
      DO 700 K=1,NSET                                                           
        LBOT=0.0                                                                
        LTOP=0.0                                                                
        IFLAG=0                                                                 
        JFLAG=0                                                                 
        KFLAG=0                                                                 
        LEBOT=0.                                                                
        LETOP=0.                                                                
        DETOP=0.                                                                
        DFTOP=0.                                                                
        DEBOT=0.                                                                
        DFBOT=0.                                                                
      WRITE(6,1) K                                                              
    1 FORMAT(///3X,'RESULTS OF DATA SET NUMBER = ',I5//)                        
      READ(5,1000) P,DROD,PITCH,TL,TIN,TSAT,HIN                                 
      READ(5,1000) MASS,DEC,MINM,LBOT,LTOP,LEBOT,LETOP                          
      READ(5,1001) NQ,IFLAG,JFLAG,KFLAG                                         
      READ(5,1000) (Q(I),I=1,NQ)                                                
      A11 = 2.*C26/(C24+C20)                                                    
      A13 = A11*(1.+C26)                                                        
      A12 = 1./A13                                                              
      A15 = 1./C23                                                              

C                                                                               
C  CALCULATION OF HYDRAULIC DIAMETER.                                           
C                                                                               
      A = PITCH**2-PI*DROD**2/4.                                                
      PW = PI*DROD                                                              
      D = 4.*A/PW                                                               
      WRITE(6,3)                                                                
    3 FORMAT(3X,'GEOMETRICAL PARAMETERS AND INLET CONDITIONS')                  
      WRITE(6,4)                                                                
    4 FORMAT(3X,'*******************************************'/)                 
      WRITE(6,5) A,PW,D,TL                                                      
    5 FORMAT(3X,'CHANNEL FLOW AREA = ',E13.6,1X,'M**2'/3X,'WETTED PERIME        
     1TER = ',E13.6,1X,'M'/3X,'HYDRAULIC DIAMETER = ',E13.6,1X,'M'/3X,'C        
     2HANNEL AXIAL LENGTH = ',E13.6,1X,'M')                                     
      WRITE(6,6) P,TIN                                                          
    6 FORMAT(3X,'INLET PRESSURE = ',F10.4,1X,'MPA'/3X,'INLET TEMPERATURE        
     1 = ',F10.4,1X,'K')                                                        
      DO 700 I=1,NQ                                                             
      WRITE(6,7)                                                                
    7 FORMAT(//3X,'******************************')                             
      WRITE(6,10) Q(I)                                                          
   10 FORMAT(3X,'HEAT FLUX = ',E13.6,1X,'MW/M**2')                              
      WRITE(6,20)                                                               
   20 FORMAT(3X,'******************************'/)                              
      MDOT = MASS                                                               
C     IF(Q(I).EQ.0.0) MMP = 31                                                  
C     IF(Q(I).GT.0.0) MMP = 30                                                  
C     DO 700 J=1,MMP                                                            
   23 DFNONB = 0.0                                                              
      DENONB = 0.0                                                              
      DANONB = 0.0                                                              
      HFO = 0.0                                                                 
      DTNONB = 0.0                                                              
      DTSUB = 0.0                                                               
      DEBB = 0.0                                                                
      DFBB = 0.0                                                                
      DABB = 0.0                                                                
      DESUB = 0.0                                                               
      DASUB = 0.0                                                               
      DFSUB = 0.0                                                               
      DTSUB=0.0                                                                 
      DTBB = 0.0                                                                
      DEVAP = 0.0                                                               
      DAVAP = 0.0                                                               
      DFVAP = 0.0                                                               
      DTVAP = 0.0                                                               
      DROP = 0.0                                                                
C                                                                               
C  FLUID PROPERTIES AT INLET.                                                   
C                                                                               
      TBA = TIN                                                                 
      CPF = CP(TBA)                                                             
      TKW = TKCOOL(TBA,P)                                                       
      UBF = UF(TBA)                                                             
C     RHOFI = RHOF(TBA)                                                         
       PP=P*1.E6                                                                
       CALL STATE(PP,TBA,TBA,ROV,RHOFI,EV,EL,TTSAT,HVS,HLS,DTSTP,               
     1 DELDP,DEVDP,DELDT,DEVDT,DRLDP,DRVDP,DRLDT,DRVDT,IOP,IERR)                
      IF(MDOT.LT. MINM) GO TO 700                                               
      IF(MDOT .LE. 0.) GO TO 700                                                
      VELI = MDOT/(RHOFI*A)                                                     
      G = VELI*RHOFI                                                            
      WRITE(6,40) MDOT,G                                                        
   40 FORMAT(/3X,'MASS FLOW RATE = ',E13.6,1X,'KG/SEC'/3X,'MASS FLUX = '        
     1,E13.6,1X,'KG/SEC-M**2')                                                  
C                                                                               
C  CALCULATION OF THE POINT OF BUBBLE DETACHMENT (ASSUMED = ZSC) USING          
C  THE SAHA AND ZUBER METHOD.                                                   
C                                                                               
      IF(Q(I).EQ.0.0) GO TO 53                                                  
      PEC = G*D*CPF/TKW                                                         
      IF(PEC.LE.70000.) DELSUB = .0022*Q(I)*1.E6*D/TKW                          
      IF(PEC.GT.70000.) DELSUB = 153.8*Q(I)*1.E6/(G*CPF)                        
      TFZSC = TSAT-DELSUB                                                       
      ZSC = D*G*CPF*(TFZSC-TIN)/(4.*Q(I)*1.E6)                                  
      WRITE(6,50) PEC,DELSUB,TFZSC,ZSC                                          
   50 FORMAT(3X,'PECLET NUMBER = ',E13.6/3X,'DELTA-T SUB = ',E13.6,1X,'K        
     1'/3X,'FLUID TEMP. AT ZSC = ',E13.6,1X,'K'/3X,'ZSC = ',E13.6,1X,'M'        
     2)                                                                         

   53 IF(ZSC.GT.TL.OR.Q(I).EQ.0.0) ZSC = TL                                     
      IF(ZSC.LT.0.0) ZSC=0.0                                                    
C                                                                               
C  NON-BOILING PRESSURE DROPS.                                                  
C       WRITE(6,121) VELI                                                       
C                                                                               
      IF(VELI.GT.0.0) RE = G*D/UBF                                              
      IF(VELI.GT.0.0) FRIC = .316/RE**.25                                       
      IF(VELI.LE.0.0) GO TO 55                                                  
      HFO = .023*TKW/D*(G*D/UBF)**.8*(UBF*CPF/TKW)**.4                          
      TBA = TIN+0.5*(TFZSC-TIN)                                                 
C     RHOBF = RHOF(TBA)                                                         
       PP=P*1.E6                                                                
       CALL STATE(PP,TBA,TBA,ROV,RHOBF,EV,EL,TTSAT,HVS,HLS,DTSDP,               
     1 DELDP,DEVDP,DELDT,DEVDT,DRLDP,DRVDP,DRLDT,DRVDT,IOP,IERR)                
   55 IF(Q(I).EQ.0.0) RHOBF=RHOFI                                               
      IF(VELI.GT.0.0) DFNONB = ZSC*FRIC*G**2/(2.*D*RHOBF)                       
      DENONB = RHOBF*9.81*ZSC                                                   
        DFBOT=(LBOT+LEBOT)*FRIC*G**2/(2.*D*RHOFI)                               
        DEBOT=RHOFI*9.81*LBOT                                                   
         DTBOT=DEBOT+DFBOT                                                      
      TBA = TFZSC                                                               
C     RHOZSC = RHOF(TBA)                                                        
      PP=P*1.E6                                                                 
      CALL STATE (PP,TBA,TBA,ROV,RHOZSC,EV,EL,TTSAT,HVS,HLS,DTSDP,              
     1 DELDP,DEVDP,DELDT,DEVDT,DRLDP,DRVDP,DRLDT,DRVDT,IQP,IERR)                
      IF(Q(I).EQ.0.0) RHOZSC=RHOFI                                              
      IF(VELI.LE.0.0) GO TO 59                                                  
      IF(ZSC.GT.0.0) DANONB=G**2*(1./RHOZSC-1./RHOFI)                           
        IF(IFLAG .GT.  0) DANONB=0.0                                            
        IF(ZSC .LT. TL) GO TO 59                                                
        DFTOP=(LTOP+LETOP)*FRIC*G**2/(2.*D*RHOBF)                               
        DETOP=RHOBF*9.81*LTOP                                                   
        DTTOP=DETOP+DFTOP                                                       
   59 DTNONB = DFNONB+DENONB+DANONB                                             
C     WRITE(6,60) RE,FRIC,RHOBF,RHOZSC,HFO                                      
   60 FORMAT(3X,'REYNOLDS NUMBER = ',E13.6/3X,'FRICTION FACTOR = ',E13.6        
     1/3X,'AV. NON-BOILING BULK FLUID DENSITY = ',F10.3,1X,'KG/M**3'/3X,        
     2'FLUID DENSITY AT ZSC = ',F10.3,1X,'KG/M**3'/3X,'COEFFICIENT OF HE        
     3AT TRANSFER = ',F10.3,1X,'W/M**2-K')                                      
      IF(ZSC.GE.TL.OR.Q(I).EQ.0.0) GO TO 625                                    
C                                                                               
C  DETERMINATION OF ZB.                                                         
C                                                                               
      TBA = TSAT                                                                
C     RHOFZB = RHOF(TBA)                                                        
C     PP=P*1.E6                                                                 
       PB=P-0.004*P                                                             
  100  PPB=PB*1.E6                                                              
      CALL STATE(PPB,TBA,TBA,ROV,RHOFZB,EV,EL,TTSAT,HVS,HLS,DTSDP,              
     1 DELDP,DEVDP,DELDT,DEVDT,DRLDP,DRVDP,DRLDT,DRVDT,IQP,IERR)                
      CALL STATE(PPB,TTSAT,TTSAT,RHOGZB,RHOFZB,EV,EL,TTTSAT,HVS,HLS,DTSD        
     1P,DELDP,DEVDP,DELDT,DEVDT,DRLDP,DRVDP,DRLDT,DRVDT,IQP,IERR)               
      HF=HLS/1.E3                                                               
       UBF=UF(TSAT)                                                             
       UBG=UG(TSAT)                                                             
C     WRITE(6,121) RHOGZB,RHOFZB,HLS                                            
C     RHOGZB = RHOG(TBA,PB)                                                     
       TBA=.5*(TSAT+TFZSC)                                                      
       RHOSUB=.5*(RHOFZB+RHOZSC)                                                
      CPF = CP(TBA)                                                             
      ZB = MDOT*(HF-HIN)/(PI*Q(I)*1.E3*DROD)                                    
      IF(ZB.GT.TL) ZB=TL                                                        
      IF(ZB.LT.0.0) ZB=0.0                                                      
      IF(ZB .LT. ZSC) ZSC=ZB                                                    
C                                                                               
C  SUBCOOLED BOILING PRESSURE DROPS.                                            
C                                                                               
      DESUB = RHOSUB*9.81*(ZB-ZSC)                                              
      IF(ZB.GT.0.0) DASUB=(G**2*(1./RHOFZB-1./RHOZSC))                          
      RE=G*D/UF(TBA)                                                            
      IF(VELI.GT.0.0) FRIC=.316/RE**.25                                         
C                                                                               
C  BRITISH UNITS USED FOR SUBCOOLED FRICTION PR. DROP CALCULATIONS.             
C                                                                               
      FPD = (G*737.73)**2*FRIC/(2.*32.2*3600.**2*D*3.281*RHOSUB*0.06243)        
      TLBSTR= (((TSAT-273.15)*1.8+32.+60.*((Q(I)*.3169E6/.1E7)**.25/EXP(        
     1PB*.145E3/900.))-.766*Q(I)*.3169E6/(HFO*.1761))-32.)*5./9.+273.15         
      PHLOMN = (1.2*((RHOFZB/RHOGZB)-1.)*.042**.824)                            

C     WRITE(6,120)                                                              
  120 FORMAT(1X,'FPD,TLBSTR,PHLOMN,FACTOR,CORFAC')                              
C     WRITE(6,121) FPD,TLBSTR,PHLOMN,FACTOR,CORFAC                              
  121 FORMAT(1X,8E13.6)                                                         
  150 IF(G*737.73.LT.7.E5) CORFAC = 1.36+.0005*PB*.145E3+.1*(G*737.73/1.        
     1E6)-.000714*PB*.145E3*(G*737.73/1.E6)                                     
      IF(G*737.73.GT.7.E5) CORFAC = 1.26-.0004*PB*.145E3+.119*(1.E6/(G*7        
     137.73))+.00028*PB*.145E3*(1.E6/(G*737.73))                                
      FSAT = PHLOMN*CORFAC+1.                                                   
      DEVID=(7.04-3.8*.042)*(1.172-1.7*PB)+3.4*PB-0.344                         
      IF(JFLAG .GT. 0) FSAT=FSAT/DEVID                                          
      FOFISO = 1.+(TBA-TLBSTR)*(FSAT-1.)/(TSAT-TLBSTR)                          
      IF (TBA .LT. TLBSTR) FOFISO = 1.-0.001*Q(I)/HFO                           
      IF (KFLAG .GT. 0) FOFISO=1                                                
      DFSUB = 47.867*FPD*(ZB-ZSC)*3.281*FOFISO                                  
        IF(ZB .LT. TL) GO TO 160                                                
        TEXIT=TIN+TL*PI*Q(I)*1.E6*DROD/(MDOT*CPF)                               
        PPB=PB*1.E6                                                             
        CALL STATE(PPB,TEXIT,TEXIT,ROV,RHOFTL,EV,EL,TTSAT,HVS,HLS,DTSDP,        
     1  DELDP,DEVDP,DELDT,DEVDT,DRLDP,DRVDP,DRLDT,DRVDT,IQP,IERR)               
C       WRITE(6,121) RHOFTL                                                     
C       WRITE(6,121) TEXIT,RHOFTL                                               
        DASUB=G**2*(1./RHOFTL-1./RHOZSC)                                        
         DESUB=.5*(RHOFTL+RHOZSC)*9.81*(TL-ZSC)                                 
        CPEXIT=CP(TEXIT)                                                        
        TKEXIT=TKCOOL(TEXIT,PB)                                                 
        UBFEX=UF(TEXIT)                                                         
        RE=G*D/UBFEX                                                            
        FRIC=.316/RE**.25                                                       
        DFTOP=(LTOP+LETOP)*FRIC*G**2/(D*RHOFTL*2.)                              
        DETOP=RHOFTL*9.81*LTOP                                                  
  160   IF(IFLAG .GT.  0) DASUB=0.0                                             
                                                                                
      DTTOP=DFTOP+DETOP                                                         
      DTSUB = DESUB+DASUB+DFSUB                                                 
C     WRITE(6,170)                                                              
C 170 FORMAT(1X,'CORFAC,FSAT,FOFISO,TBA,TLBSTR')                                
C     WRITE(6,121)CORFAC,FSAT,FOFISO,TBA,TLBSTR                                 
C                                                                               
C  ITERATIONS OF PRESSURE AT ZB.                                                
C                                                                               
      PBB = (P*1.E6-(DTNONB+DTSUB+DTBOT))                                       
C     WRITE(6,121) PB,PBB                                                       
      PBITE = ABS((PBB/1.E6-PB)/PB)                                             
      IF(PBITE.GT.0.00005) PB=PBB/1.E6                                          
      IF(PBITE.GT.0.00005) GO TO 100                                            
        IF (KFLAG .LE. 0) GO TO 162                                             
        DANONB=DANONB+DASUB                                                     
        TBA=(TIN+TSAT)/2.                                                       
                                                                                
        IF(ZB .GE. TL) TBA=(TEXIT+TIN)/2.                                       
        PPB=PB*1.E6                                                             
        CALL STATE(PPB,TBA,TBA,ROV,RHOFBA,EV,EL,TTSAT,HVS,HLS,DTSDP,            
     1  DELDP,DEVDP,DELDT,DEVDT,DRLDP,DRVDP,DRLDT,DRVDT,IQP,IERR)               
        DENONB=RHOFBA*9.81*ZB                                                   
         DFNONB=DFNONB*(RHOBF/RHOFBA)*(ZB/ZSC)                                  
        DTNONB=DANONB+DENONB+DFNONB                                             
        DESUB=0.0                                                               
        DASUB=0.0                                                               
         DFSUB=0.0                                                              
        DTSUB=0.0                                                               
  162 WRITE(6,200) ZB                                                           
  200 FORMAT(3X,'ZB = ',F10.4,1X,'M')                                           
                                                                                
      IF(ZB.GE.TL) GO TO 625                                                    
C                                                                               
C  DETERMINATION OF ZV.                                                         
C                                                                               
      PV = (P-0.006*P)                                                          
C 300 HG = ENTHAG(PV)                                                           
  300   PPV=PV*1.0E6                                                            
        CALL STATE(PPV,TSAT,TSAT,ROV,ROL,EV,EL,TTSAT,HVS,HLS,DTSTP,             
     1DELDP,DEVDP,DELDT,DEVDT,DRLDP,DRVDP,DRLDT,DRVDT,IOP,IERR)                 
        HG=HVS/1.E3                                                             
      ZV = MDOT*(HG-HIN)/(PI*Q(I)*1.E3*DROD)                                    
      IF(ZV.GT.TL) ZV=TL                                                        
      IF(ZV.LT.0.0) ZV=0.0                                                      
C                                                                               
C  CALCULATION OF MARTINELLI-NELSON VOID FRACTION.                              

C                                                                               
      XZSC = -CPF*DELSUB/((HG-HF)*1.E3)                                         
      XZB = -XZSC/2.71828                                                       
      XTTZB = (UBF/UBG)**.1*(((1.-XZB)/XZB)**.9)*(RHOGZB/RHOFZB)**.5            
      RGZB = 1.-XTTZB/((XTTZB**2+20.*XTTZB+1.)**.5)                             
C                                                                               
C   DETERMINATION OF X(Z=TL)=XOUT                                               
C                                                                               
C     HGPB=ENTHAG(PB)                                                           
        PPB=PB*1.E6                                                             
        CALL STATE (PPB,TSAT,TSAT,ROV,ROL,EV,EL,TTSAT,HVS,HLS,DTSDP,            
     1  DELDP,DEVDP,DELDT,DEVDT,DRLDP,DRVDP,DRLDT,DRVDT,IQP,IERR)               
        HGPB=HVS/1.E3                                                           
C       WRITE(6,121) HVS                                                        
      HOUT=TL*(PI*Q(I)*1.E3*DROD)/MDOT+HIN                                      
      XEOUT=(HOUT-HF)/(HGPB-HF)                                                 
      XOUT=XEOUT-XZSC*EXP(XEOUT/XZSC-1.)                                        
      IF(XEOUT .GT. 1.0) XEOUT = 1.0                                            
      IF(XOUT  .GT. 1.0) XOUT  =1.0                                             
C                                                                               
C  BULK BOILING PRESSURE DROPS.                                                 
C                                                                               
      XZV = 1.-XZSC*EXP(1./XZSC-1.)                                             
      IF(ZV .GE. TL) XZV=XOUT                                                   
      IF(XZV.GT.1.0) XZV=1.0                                                    
      XTTZV = (UBF/UBG)**.1*(((1.-XZV)/XZV)**.9)*(RHOGZB/RHOFZB)**.5            
      RGZV = 1.-XTTZV/((XTTZV**2+20.*XTTZV+1.)**.5)                             
      RHOB = RHOFZB-0.5*(RGZB+RGZV)*(RHOFZB-RHOGZB)                             
      DEBB = RHOB*9.81*(ZV-ZB)                                                  
        RHOZV=RHOFZB-RGZV*(RHOFZB-RHOGZB)                                       
      DZV = XZV**2/(RHOGZB*RGZV)                                                
      IF(XZV.NE.1..OR.RGZV.NE.1.) DZV = (1.-XZV)**2/(RHOFZB*(1.-RGZV)           
     1)+DZV                                                                     
      DZB = (1.-XZB)**2/(RHOFZB*(1.-RGZB))+XZB**2/(RHOGZB*RGZB)                 
      IF(ZV.GT.0.0) DABB = (G**2)*(DZV-DZB)                                     
        XEZV=1.0                                                                
         IF(ZV .GE. TL) XEZV=XEOUT                                              
      PHLOMN=1.2*((RHOFZB/RHOGZB)-1.)*(XEZV)**.824                              
      DFBB=(G*737.73)**2*FRIC*(PHLOMN*CORFAC/1.824+1.)*(ZV-ZB)/(RHOFZB*         
     10.06243*2*32.2*3600.**2*D)*47.867                                         
       DEVID=((7.04-3.8*XEZV)*(1.172-1.7*PB)+3.4*                               
     1 PB-.344)                                                                 
        IF(IFLAG .GT.  0) DABB=0.                                               
        IF(ZV .LT. TL ) GO TO 350                                               
        DFTOP=G**2*FRIC*(PHLOMN*CORFAC+1.0)*(LTOP+LETOP)/                       
     1  (RHOFZB*D*2.)                                                           
        IF(JFLAG .GT. 0) DFTOP=DFTOP/DEVID                                      
        DETOP=RHOZV*9.81*LTOP                                                   
        DTTOP=DFTOP+DETOP                                                       
  350   IF(JFLAG .GT.  0) DFBB=DFBB/DEVID                                       
C     WRITE (6,320)                                                             
  320 FORMAT(1X,'DFBB,FRIC,PHLOMN,CORFAC,ZB,RHOFZB,RHOGZB')                     
C     WRITE(6,121) DFBB,FRIC,PHLOMN,CORFAC,ZB,RHOFZB,RHOGZB                     
      DTBB = DEBB+DABB+DFBB                                                     
C                                                                               
C  ITERATIONS OF PRESSURE AT ZV.                                                
C                                                                               
      PVV = (P*1.E6-(DTNONB+DTSUB+DTBB+DTBOT))                                  
      PVITE = ABS((PVV/1.E6-PV)/PV)                                             
      IF(PVITE.GT.0.00005) PV=PVV/1.E6                                          
      IF(PVITE.GT.0.00005) GO TO 300                                            
      WRITE(6,400) ZV,XZSC,XZB,RGZB,XZV,RGZV,XEOUT,XOUT                         
  400 FORMAT(3X,'ZV= ',F10.4,1X,'M'/                                            
     1 3X,'EQUILIBRIUM QUALITY AT ZSC = ',E13.6/                                
     2 3X,'NON EQUILIBRIUM QUALITY AT ZB =',E13.6/                              
     3 3X,'MARTINELLI-NELSON VOID FRACTION AT ZB =',F10.4/                      
     4 3X,'NON EQUILIBRIUM QUALITY AT ZV = ',E13.6/                             
     5 3X,'MARTINELLI-NELSON VOID FRACTION AT ZV =', E13.6/                     
     6 3X,'QUALITY AT CHANNEL OUTLET , EQ = ',F10.4,' NON EQ = ',F10.4)         
      IF(ZV.GE.TL) GO TO 625                                                    
C                                                                               
C  DETERMINATION OF VAPOR DENSITY AT EXIT.                                      
C                                                                               
       HTL = Q(I)*1.E3*PI*DROD*TL/MDOT+HIN                                      
      TG = TSAT+2.                                                              
      IF(PV*1.E6.GT.2.E6) GO TO 500                                             
      ES = C6+C7*(1./(C8+PV*1.E6))                                              
      GS = C9+(C11*PV*1.E6+C10)*PV*1.E6                                         
      GO TO 510                                                                 

  500 ES = C12+(C14*PV*1.E6+C13)*PV*1.E6                                        
      GS = C15+(C17*PV*1.E6+C16)*PV*1.E6                                        
  510 CPS = C4*(1.0-A15*TSAT)**C5                                               
      BETA = TSAT**2*(1.0-1.0/(A11*CPS-1.0)**2)                                 
  520 EG = ES+A12*((TG-TSAT)+(TG**2-BETA)**.5-TSAT/(A11*CPS-1.0))               
      DENG = PV*1.E6/((GS-1.0)*ES+C26*(EG-ES))                                  
      HG = EG+PV*1.E6/DENG                                                      
      HABS = ABS((HG/1.E3-HTL)/HTL)                                             
      IF(TG.GT.700.) GO TO 530                                                  
      IF(HABS.GT.0.005) TG = TG+1.                                              
      IF(HABS.GT.0.005) GO TO 520                                               
C                                                                               
C  SINGLE PHASE VAPOR PRESSURE DROPS. THESE PRESSURE DROPS MAY NOT BE           
C  COMPATIBLE WITH THOSE CALCULATED EARLIER AND ARE NOT TABULATED.              
C                                                                               
  530 TBA = TG-0.5*(TG-TSAT)                                                    
      IF(TBA.GT.640.) TBA = 640.                                                
      IF(TBA.GT.640.) WRITE(6,570)                                              
  570 FORMAT(3X,'THE AV. VAPOR TEMP. HAS EXCEEDED THE RANGE OF VALIDITY         
     1OF THE VAPOR VISCOSITY CORRELATION. SO AS TO AVOID A SUBSEQUENT NE        
     2GATIVE ARGUMENT,TBA IS TAKEN TO BE 640. K')                               
      UBG = UG(TBA)                                                             
      RE = G*D/UBG                                                              
C     FRIC = .316/RE**.25                                                       
      DEVAP = 0.5*(RHOGZB+DENG)*9.81*(TL-ZV)                                    
      DAVAP = G**2*(1./DENG-1./RHOGZB)                                          
C     DFVAP = FRIC*G**2*(TL-ZV)/(2.*D*0.5*(RHOGZB+DENG))                        
       PHOLMN=1.2*(RHOFZB/RHOGZB-1.)                                            
       DFVAP=FRIC*G**2.*(TL-ZV)*(PHOLMN*CORFAC+1.)/(RHOFZB                      
     1 *D*2.)                                                                   
       DFTOP=FRIC*G**2.*(LTOP+LETOP)*(PHOLMN*CORFAC+1.)/(RHOFZB*D*2.)           
      IF(IFLAG .GT.  0) DAVAP=0.0                                               
C     DFVAP = DFVAP+FRIC*G**2*(LTOP+LETOP)/(2.*D*DENG)                          
      DETOP=DENG*9.81*LTOP                                                      
       IF(JFLAG .LE. 0) GO TO 580                                               
       DFVAP=DFVAP/DEVID                                                        
       DFTOP=DFTOP/DEVID                                                        
  580 DTTOP=DETOP+DFTOP                                                         
      DTVAP = DEVAP+DAVAP+DFVAP                                                 
      WRITE(6,600) TG,HTL,DENG,HG,RE,DEVAP,DAVAP,DFVAP,DTVAP                    
  600 FORMAT(3X,'VAPOR TEMPERATURE AT EXIT = ',F10.3,1X,'K'/3X,'ENTHALPY        
     1 AT EXIT(HEAT BALANCE)= ',F10.1,1X,'KJ/KG'/3X,'VAPOR DENSITY AT EX        
     2IT = ',F10.3,1X,'KG/M**3'/3X,'ENTHALPY AT EXIT(CORRELATED) = ',F10        
     3.1,1X,'J/KG'/3X,'VAPOR REYNOLDS NUMBER = ',F10.1/3X,'ELEVATION VAP        
     4OR PR. DROP = ',F10.3,1X,'PA'/3X,'ACCELERATION VAPOR PR. DROP = ',        
     5F10.3,1X,'PA'/3X,'FRICTION VAPOR PR. DROP = ',F10.3,1X,'PA'/3X,'TO        
     6TAL VAPOR PR. DROP = ',F10.2,1X,'PA')                                     
C                                                                               
C  TOTAL PRESSURE DROP.                                                         
C                                                                               
  625 DROP = DTNONB+DTSUB+DTBB+DTVAP+DTTOP+DEBOT+DFBOT                          
      WRITE(6,650) DROP                                                         
  650 FORMAT(/3X,'TOTAL PRESSURE DROP = ',F10.0,1X,'PA')                        
C                                                                               
C  TABULATION OF PRESSURE DROP COMPONENTS                                       
C                                                                               
      DF = DFNONB+DFSUB+DFBB+DFTOP+DFBOT                                        
      DE = DENONB+DESUB+DEBB+DETOP+DEBOT                                        
      DA = DANONB+DASUB+DABB                                                    
      DFCENT = DF/DROP*100.                                                     
      DECENT = DE/DROP*100.                                                     
      DACENT = DA/DROP*100.                                                     
      DTNCEN = DTNONB/DROP*100.                                                 
      DTSCEN = DTSUB/DROP*100.                                                  
      DTBCEN = DTBB/DROP*100.                                                   
      WRITE(6,651)                                                              
  651 FORMAT(/3X,'******** TABULATION OF PRESSURE DROP COMPONENTS IN PAS        
     1CALS *********')                                                          
      WRITE(6,652)                                                              
  652 FORMAT(3X,'********************************************************       
     1*************')                                                           
      WRITE(6,653)                                                              
  653 FORMAT(13X,'BOILING REGIME                     SUBTOTAL (COMP./TOT        
     1AL)%')                                                                    
      WRITE(6,654)                                                              
  654 FORMAT(3X,'COMPONENT NON-BOILING  SUBCOOLED      BULK')                   
      WRITE(6,655)                                                              
  655 FORMAT(3X,'********************************************************       
     1*************')                                                           

      WRITE(6,656) DFNONB,DFSUB,DFBB,DF,DFCENT                                  
  656 FORMAT(3X,'FRICTION',4X,F10.0,1X,3F10.0,8X,F5.2)                          
      WRITE(6,657) DENONB,DESUB,DEBB,DE,DECENT                                  
  657 FORMAT(3X,'ELEVATION',3X,F10.0,1X,3F10.0,8X,F5.2)                         
      WRITE(6,658) DANONB,DASUB,DABB,DA,DACENT                                  
  658 FORMAT(3X,'ACCELERATION',F10.0,1X,3F10.0,8X,F5.2)                         
      WRITE(6,659)                                                              
  659 FORMAT(3X,'*******************************************************        
     1*************')                                                           
      WRITE(6,660)                                                              
  660 FORMAT(50X,'TOTAL')                                                       
      WRITE(6,661) DTNONB,DTSUB,DTBB,DROP                                       
  661 FORMAT(3X,'SUBTOTAL',4X,F10.0,1X,3F10.0)                                  
      WRITE(6,662)                                                              
  662 FORMAT(46X,'**********')                                                  
      WRITE(6,663) DTNCEN,DTSCEN,DTBCEN                                         
  663 FORMAT(3X,'(REGIME/TOTAL)%',1X,F6.2,5X,F6.2,4X,F6.2)                      
      WRITE(6,675)                                                              
  675 FORMAT(/3X,'******************************************************        
     1**************')                                                          
      WRITE(6,676) DETOP,DFTOP,DEBOT,DFBOT                                      
  676  FORMAT(3X,'TOP    UNHEATED PR. DROP  ELEV  =',F8.2,' FRI=',F8.2/         
     1 3X,'BOTTOM UNHEATED PR. DROP  ELEV  =',F8.2,' FRI=',F8.2)                
      MDOT=MDOT-DEC                                                             
      GO TO 23                                                                  
  700 CONTINUE                                                                  
 1000 FORMAT(7f10.0)                                                                 
 1001 FORMAT(6i5)                                                                 
      STOP                                                                      
      END                                                                       

